

Integration in polar coordinates

Worksheet by Mike May, S.J. - maymk@slu.edu


Revised by Russell Blyth - blythrd@slu.edu


> **restart:**

▼ A review of plotting in polar coordinates

The first task in computing double integrals using polar coordinates is to correctly sketch graphs of functions described in polar coordinates. On a calculator you switch to polar mode. With Maple you use the `coord=polar` option in the plot command. You can plot the curve either with r as a function of θ , or with both r and θ described as functions of a parameter t . When we plot in polar coordinates it is usually wise to use the `scaling=CONSTRAINED` option so the axes have the same scale.


```
> plot(cos(2*theta),theta=0..2*Pi, coords=polar, scaling=
CONSTRAINED);
plot([2+sin(2*t),Pi*sin(t),t=0..Pi], coords=polar, scaling=
CONSTRAINED);
```






To plot several polar curves together at the same time we use set notation, just like we did in Cartesian coordinates.

```
> plot({1,2*sin(theta)}, theta=0..2*Pi, coords=polar, scaling=
CONSTRAINED);
```



Plotting several curves together allows us to plot regions made of a number of curves. The region plotted below is the portion of an annulus between two specified angles. The fifth (empty) curve is included so that Maple includes the origin in the plot (the t -range for this "curve" can be any nonempty interval).

```
> plot({[1+t, Pi/6, t=0..1], [1+t, Pi/3, t=0..1], [1, t, t=Pi/6..Pi/3],
        [2, t, t=Pi/6..Pi/3], [0, 0, t=0..2*Pi]}, coords=polar, scaling=
        CONSTRAINED);
```



Exercise

1. Plot the curves $r = 3 \cos(\theta)$ and $r = 1 + \cos(\theta)$ on the same graph. Find the points of intersection. (You may want to use either your calculator or paper and pencil to find the points of intersection.)

>

Once we can sketch curves in polar coordinates the process of setting up an integral in polar coordinates is similar to the process involved in setting up a double integral in Cartesian coordinates. The biggest challenge is finding the correct limits of integration. We will also discuss switching the order of integration.

Finding the limits of integration

Setting up $drd\theta$ integrals

Consider first the case of integrals using the order of integration $drd\theta$. Since dr is on the inside we have a region bounded by curves $r = g(\theta)$ and $r = h(\theta)$, with the value of θ bound by two constant angles. The integration with respect to r for a particular θ is along a radial line. The following block of code is designed to help you visualize what the limits of integration mean. The curves $r = g(\theta)$ and $r = h(\theta)$ are specified in the fourth and fifth lines as `lowr` and `highr`

(the inside and outside curves, respectively).

```
> r:='r':  theta := 'theta':
lowtheta  := Pi/6;
hightheta := 3*Pi/2;
lowr      := theta -> 1.5-sin(theta);
highr     := theta -> 3 + cos(theta);
print(`Region of integration for ` ,
      Int(Int(f(r, theta)*r,r=lowr(theta)..highr(theta)),
        theta=lowtheta..hightheta));
inside := plot([lowr(theta), theta,theta=lowtheta..hightheta]
,
              color=red, coords=polar):
outside := plot([highr(theta), theta,theta=lowtheta..
hightheta],
              color=green, coords=polar):
line := {}:
for i from 0 to 10 do
  tval := evalf(lowtheta + i/10*(hightheta-lowtheta)):
  if (abs(evalf(lowr(tval)) - highr(tval)))>0) then
    line := line union
          {[r,tval, r=evalf(lowr(tval))..evalf(highr(tval))]}
;
  end if
end do:
plotlines := plot(line,coords=polar, color=BLACK) :
plots[display]({inside, outside, plotlines},scaling=
CONSTRAINED);
```


$$lowtheta := \frac{1}{6} \pi$$

$$hightheta := \frac{3}{2} \pi$$

$$lowr := \theta \rightarrow 1.5 - \sin(\theta)$$

$$highr := \theta \rightarrow 3 + \cos(\theta)$$

$$\text{Region of integration for, } \int_{\frac{1}{6} \pi}^{\frac{3}{2} \pi} \int_{1.5 - \sin(\theta)}^{3 + \cos(\theta)} f(r, \theta) r \, dr \, d\theta$$



Note that since we integrate with respect to r first, the r -limits are functions of θ while the θ limits are constants. We first integrate from the inside (red) curve to the outside (green) curve, along the radial lines, and then add up (integrate) over the range of values of θ .

▼ Exercises

2. Find the limits of integration to integrate over the region inside the curve $r = 1 + \cos(\theta)$ and outside the curve $r = 1$. Modify the code above to show that you have the correct region.

>

3. Find the limits of integration to integrate over the region inside the curve $r = 3 \cos(\theta)$ and outside the curve $r = 1 + \cos(\theta)$. Modify the code above to show that you have the correct region.

>

▼ Setting up $d\theta dr$ integrals

Similarly we can set up integrals using the order of integration $dr d\theta$. Now the region of integration is bounded by curves $\theta = g(r)$ and $\theta = h(r)$ with the value of r being bound by two constants. Instead of integrating first on radial lines, we start by integrating along circular arcs with a fixed values of r . The curves $\theta = g(r)$ and $\theta = h(r)$ appear in the fourth and fifth lines of the code as `lowtheta` and `hightheta`. Remember that angles increase in a counterclockwise

direction, so lowtheta is the angle on the clockwise extreme of the region. You need to take care that lowtheta is a smaller angle than hightheta (sometimes this means that lowtheta will be a negative angle).

```

> r:='r':  theta := 'theta':
  lowr  := 1;
  highr := 5;
  lowtheta := r -> Pi*r/6;
  hightheta := r -> Pi*(2-r/12);
  print(`Region of integration for `, Int(Int(f(r, theta)*r,
      theta=lowtheta(r)..hightheta(r)),r=lowr..highr))
;
  lowthetacurve := plot ([r,lowtheta(r), r=lowr..highr],
      color=red, coords=polar) :
  highthetacurve := plot ([r, hightheta(r), r=lowr..highr],
      color=green, coords=polar) :
  arcs := {} :
    for i from 0 to 10 do
      tempr := evalf(lowr + i/10*(highr-
  lowr)):
      if (abs(evalf(lowtheta(tempr)-hightheta(tempr)))>0) then
        arcs := arcs union {[tempr, theta,
          theta=lowtheta(tempr)..hightheta(tempr)]}:
      end if:
    end do:
  grapharcs := plot(arcs,coords=polar, color=BLACK) :
  plots[display] ( {lowthetacurve, highthetacurve, grapharcs }
    ,scaling=CONSTRAINED) ;

```


lowr := 1

highr := 5

$$\text{lowtheta} := r \rightarrow \frac{1}{6} \pi r$$

$$\text{hightheta} := r \rightarrow \pi \left(2 - \frac{1}{12} r \right)$$

$$\text{Region of integration for } \int_1^5 \int_{\frac{1}{6} \pi r}^{\pi \left(2 - \frac{1}{12} r \right)} f(r, \theta) r d\theta dr$$



Exercises

4. Find the limits of integration to integrate over the region inside both curves $r = 1 + \cos(\theta)$ and $r = 1$. Modify the code above to show that you have the correct region. Explain why you would use `dthetadr` rather than `drdtheta` for this problem.


```
> r:='r': theta := 'theta':
  lowr := 0;
  highr := 1;
  lowtheta := r -> -arccos(r-1);
  hightheta := r -> arccos(r-1);
  print(`Region of integration for `, Int(Int(f(r, theta)*r,
    theta=lowtheta(r)..hightheta(r)), r=lowr..
  highr));
  lowthetacurve := plot ([r, lowtheta(r), r=lowr..highr],
    color=red, coords=polar) :
  highthetacurve := plot ([r, hightheta(r), r=lowr..highr],
    color=green, coords=polar) :
  arcs := {} :
  for i from 0 to 10 do
    tempr := evalf(lowr + i/10*(highr-
  lowr)):
    if (abs(evalf(lowtheta(tempr)-hightheta(tempr)))>0) then
```

```

arcs := arcs union {[tempr, theta,
  theta=lowtheta(tempr)..hightheta(tempr)]}:
end if:
end do:
grapharcs := plot(arcs,coords=polar, color=BLACK) :
plots[display] ( {lowthetacurve, highthetacurve, grapharcs
}
,scaling=CONSTRAINED) ;
lowr := 0
highr := 1
lowtheta := r → -arccos(r-1)
hightheta := r → arccos(r-1)

```

Region of integration for $\int_0^1 \int_{-\arccos(r-1)}^{\arccos(r-1)} f(r, \theta) r \, d\theta \, dr$



5. Find the limits of integration to integrate over the region inside both curves $r = 1 + \cos(\theta)$ and $r = 1$. Modify the code above to show that you have the correct region. Explain why you would use $d\theta dr$ rather than $dr d\theta$ for this problem.

>

▼ **Changing order of integration**

Some regions can be described to use either $rdrd\theta$ or $d\theta dr$. When we switch the order of integration we need to change the limits as well.

Exercise

6. The region inside the curve $r=1$ and outside the curve $r = 1 + \cos(\theta)$ can be set up in either order. Find the limits of integration both ways. Show that you have the correct region.

>

Integrating over a region in polar coordinates

Recall that in polar coordinates, dA is $rdrd\theta$ or $r d\theta dr$. Thus, for example, to find the area of the

region bounded by $r=1$, $r=2$, $\theta = \frac{\pi}{6}$, and $\theta = \frac{\pi}{3}$, we evaluate the integral $\int_{\frac{\pi}{6}}^{\frac{\pi}{3}} \int_1^2 r dr d\theta$.

> `Int(Int(r, r=1..2), theta=Pi/6..Pi/3)=int(int(r, r=1..2), theta=Pi/6..Pi/3);`

$$\int_{\frac{1}{6}\pi}^{\frac{1}{3}\pi} \int_1^2 r dr d\theta = \frac{1}{4} \pi$$

Exercises

7. Find the area inside both curves $r=1$ and $r = 1 + \cos(\theta)$.

>

8. Integrate the function $\sin(r^2)$ over the disk of radius 2 centered at the origin.

>

Changing coordinate systems

One of the reasons we want to know how to integrate in both Cartesian and polar coordinates is that some integrals work out nicely in one coordinate system and are ugly or impossible in another. To change an integral in Cartesian coordinates into polar coordinates, we need to do several things.

First sketch the region with its boundary curves. Then convert the formulas of the boundary curves, the function to be integrated and dA into polar form. We are then ready to set up the integral and integrate.

Consider the integral $\int_{-b}^b \int_{-\sqrt{b^2-x^2}}^{\sqrt{b^2-x^2}} \frac{1}{a^2+x^2+y^2} dy dx$.

> `int(int(1/(a^2+x^2+y^2), y=-sqrt(b^2-x^2)..sqrt(b^2-x^2)), x=-b..b);`

$$\int_{-b}^b \frac{2 \arctan\left(\frac{\sqrt{b^2-x^2}}{\sqrt{a^2+x^2}}\right)}{\sqrt{a^2+x^2}} dx$$

Depending on the version of Maple you are using, it either chokes on this integral, or gives an answer involving functions we don't know how to evaluate. However the integral above converts to

$$\int_0^b \int_0^{2\pi} \frac{r}{a^2+r^2} d\theta dr$$

in polar form, which can easily be computed by hand using the substitution

$u = r^2$. Maple also has no problem with it, provided we tell it that a is to be regarded as a real value (ignore the \sim that appears after the a in the answer - that is Maple's way of reminding us that we have placed a restriction on a).

```
> assume(a, real);
int(int(r/(a^2+r^2), theta=0..2*Pi), r=0..b);
      -ln(a~^2) pi+pi ln(a~^2+b^2)
```

Exercise

9. Convert the integral $\int_0^{\sqrt{2}} \int_y^{\sqrt{4-y^2}} x y dx dy$ to polar form and evaluate.

>

Extra credit

At the beginning of this worksheet we plotted a region described by the parametric curve $[2+\sin(2t), \pi\sin(t), t=0..\pi]$. We want to find the area of the region bounded by this curve to 3 decimal places. Find the area of the region and cleanly write up your work carefully justifying your method.

>

>